/*----------------------------------------------------
*                                                   *
*                                                   *
*  University of Iowa : SAS User's Group            *
*  Useful Tips Code                                 *
*  Last Modified : 12Aug2013                        *
*  Programmer : JJB                                 *
*                                                   *
-----------------------------------------------------
*                                                   *
*  List of examples:                                *
*                                                   *
*  1) Using cards to make dataset                   *
*  2) Sorting by using natural "language" rules     *
*  3) Renaming variables                            *
*  4) Counting values that do not exist             *
*  5) Combining summary data with individual values *
*  6) Attributes of your data are unknown           *
*  7) Coding patterns of vars. into one variable    *
*  8) Keeping only the desired output               *
*  9) Adding footnotes and usernames to output      *
*                                                   *
---------------------------------------------------*/




/****************************************************/
/****************************************************/
*****  problem : including data in a program using cards

data measures;
input f_name $10. sex $  age height   weight ;
cards;
Alfred      M      14     69.0      112.5
Alice       F      13     56.5       84.0
Barbara     F      13     65.3       98.0
Ronald      M      15     67.0      133.0
Thomas      M      11     57.5       85.0
William     M      15     66.5      112.0
run;



/****************************************************/
/****************************************************/
*****  problem : sorting cases by using "language" rules


data bf_sort;
input parts $12.;
cards;
10.2.f.ii
10.2.a.iii
10.2.f.i
10.2
8.4.5.b.i
10.2.c.iii
10.2.d.i
10.2.d.ii
8.4.5.c.i
10.2.e.ii
10.2.e.i
7.5.2.b
7.5.2.c
8.4.5.e.i
10.2.b.iii
run;


/****************************************************/
/****************************************************/
*****  problem : renaming variables

data toons;
input ID $4. yakko wakko dot buttons mindy;
cards;
0001 0 1 1 2 3
0002 2 2 1 2 0
0003 3 4 4 3 2
0004 1 0 0 2 1
0005 4 3 4 4 4
;
run;


*****  keep and order the variables as desired ;

data toons_reorder; 
retain id yakko dot mindy;
set toons;
keep id yakko dot mindy ;
run;

*****  rename the variables using an array  ;

data toons_renamed (keep=ID toon1 -- toon3);
	set toons_reorder;
	
	array old{3} yakko -- mindy; 
	array toon{3} ;

	do i=1 to 3;
		toon[i] = old[i];
	end;
run;	

*****  using the colon ":" operator...

data tmp ; set toons_renamed;
toon_sum = sum(of toon: ) ;
run;

proc means data = tmp;
var toon: ;
run;


/****************************************************/
/****************************************************/
*****  problem : counting values that do not exist

proc format;
      value race  1 = 'American Indian/Alaska Native'
                  2 = 'Asian'
                  3 = 'Native Hawaiian or Other Pacific Islander'
                  4 = 'Black or African American'
                  5 = 'White'
                  6 = 'More Than One Race'
                  7 = 'Unknown or Not Reported';
 
      value eth   1 = 'Hispanic or Latino'
                  2 = 'Not Hispanic or Latino'
                  3 = 'Unknown or Not Reported';
 
      value sex   1 = 'Male'
                  2 = 'Female'
                  3 = 'Unknown or Not Reported';
run;

data race;
input race  eth  sex   ;
cards;
1   2   1
1   2   1
1   2   2
2   2   1
2   2   2
2   2   1
3   2   2
3   2   2
3   2   1
4   2   2
4   2   1
4   2   2
run;

***** Does not work correctly ;

proc tabulate data = race;

class eth sex ;
format eth eth. sex sex.  ; 
table (eth all = 'Ethnicity Category: Total of All Subjects'), 
sex all = 'Total'
 
            / printmiss misstext='0';
 
      title 'No Format - Hispanic is Missing';
run;

***** Does work correctly ;

proc tabulate data = race;
format eth eth. sex sex.  ;
class eth sex / preloadfmt;
 
table (eth all = 'Ethnicity Category: Total of All Subjects'), 
sex all = 'Total'
 
            / printmiss misstext='0';
 
      title 'Formatted - Hispanic is Present';
run;





/****************************************************/
/****************************************************/
*****  problem : combining summary data with individual records


data prop_scores;

     do i = 1 to 30;
           pred_prob = ranuni(345343);
           output;
     end;

     keep pred_prob;
run;


/*  find quintiles of scores  */

proc univariate data=prop_scores;
   var pred_prob;
   output out=Pctls pctlpts  = 20 40 60 80 
                    pctlpre  = prop_
                    pctlname = pct20 pct40 pct60 pct80;
run;



/*  add quintiles to data set with scores  */

data prop_scores2;
     if _n_ = 1 then set pctls;
     set prop_scores;
run;


/*  create variable prop_quint that                */
/*  categorizes propensity scores into quintiles.  */


data prop_scores2;
     set prop_scores2;

     if missing(pred_prob) then prop_quint = .;
     
         if          . < pred_prob <= prop_pct20 then prop_quint = 1;
         if prop_pct20 < pred_prob <= prop_pct40 then prop_quint = 2;
         if prop_pct40 < pred_prob <= prop_pct60 then prop_quint = 3;
         if prop_pct60 < pred_prob <= prop_pct80 then prop_quint = 4;
         if prop_pct80 < pred_prob               then prop_quint = 5;
run;




/****************************************************/
/****************************************************/
*****  problem : knowing your data

proc sql;
  create table AllVars as
    select *
      from dictionary.columns
      
	  where libname = 'SASHELP' and memname = 'CLASS' ; 
      
      ***** !!!!! note that the libname and memname above  must be all upper-case ;
quit;



proc print data=AllVars; 
run;




/****************************************************/
/****************************************************/
*****  problem : combining several variables into one variable 



data demos;
input  id  ethnicity  raceai racea  raceaa racenh racew raceunk racenr ;
cards;
1  2 0 0 0 0 1 0 0
2  2 0 0 0 0 1 0 0
3  2 0 0 1 0 1 0 0
4  2 0 1 0 0 0 0 0
5  1 0 0 1 0 0 0 0
6  2 1 0 0 0 0 0 0
7  1 0 0 0 0 0 1 0
8  1 0 0 0 0 1 0 0
9  1 0 0 0 0 0 0 1
10 3 0 1 0 1 1 0 0
run;

data race; set demos;

*****  recode ethnicity to group together Unknown and Not Reported  ;

if ethnicity = 3 or ethnicity = 4 then rep_ethnicity = 3;
else rep_ethnicity = ethnicity;

format rc_race $char5. rc_unk $char2. ;


***** this "recodes" the race variables into one variable of length 5 made up of 0's and 1's  ;

rc_race = cats(raceai, racea, raceaa, racenh, racew);
rc_unk =  cats(raceunk, racenr);

***** this gives the count of the number of times a 1 occurs in:  rc_race and rc_unk  ;

c_rc_race = count(rc_race, "1") ;   *  counts the number of times a 1 occurs in rc_race ;
c_rc_unk  = count(rc_unk , "1") ;   *  counts the number of times a 1 occurs in rc_unk  ;


***** this gives the race category that matches the pattern  ;

if rc_race = "10000" then rep_race = 1;   *  i.e., american indian / AK native ! and nothing else  ;
if rc_race = "01000" then rep_race = 2;   *  i.e., asian  ! and nothing else  ;
if rc_race = "00100" then rep_race = 3;   *  i.e., african american ! and nothing else ;
if rc_race = "00010" then rep_race = 4;   *  i.e., native hawaiian ! and nothing else  ;
if rc_race = "00001" then rep_race = 5;   *  i.e., white ! and nothing else  ;


*  if the count of 1's is greater than 1, the person is multi-racial  ;
if c_rc_race > 1 then rep_race = 6;       *  i.e., more than 1 race ;

*  if the count of 1's is greater than 0, the person's race is unknown or not reported ;
if c_rc_unk > 0 then rep_race = 7;        *  i.e., unknown or not reported ;   

label 	rep_race = "Racial Categories"
		rep_ethnicity = "Ethnic Category"  ;

run;



proc format;

    value   frep_race       1 = "American Indian / AK Native"
                            2 = "Asian"
                            3 = "Black or African American"
                            4 = "Native Hawaiian or Other PI"
                            5 = "White"
                            6 = "More than one race"
                            7 = "Unknown or not reported"  ;

    value   frep_ethnicity  1 = "Hispanic or Latino"
                            2 = "Not Hispanic or Latino"
                            3 = "Unknown or Not reported"  ;

run;



proc freq data = race;
format rep_race frep_race. ;
tables rep_race ;
run ;



/****************************************************/
/****************************************************/
*****  problem : keeping the output you want


data selection;
input weightloss $ surv_month   cens @@;
datalines;
No  1   1    No  11  1    Yes 15  1
No  38  0    No  6   0    Yes 13  0
No  9   0    No  9   1    Yes 14  1
No  9   1    No  4   1    Yes 71  0
No  13  1    No  20  1    Yes 37  0
No  20  1    No  44  1    Yes 17  1
No  8   0    No  81  0    Yes 32  0
No  4   1    No  74  0    Yes 17  1
No  27  1    No  13  1    Yes 13  1
No  62  1    Yes 6   1    Yes 27  1
No  12  1    Yes 39  1    Yes 8   1
No  32  1    Yes 100 1    Yes 63  1
No  17  1    Yes 30  0    Yes 22  1
No  15  1    Yes 1   1    Yes 5   1
No  15  1    Yes 2   1    Yes 62  0
No  8   0    Yes 2   0    Yes 9   1
No  22  1    Yes 6   1    Yes 15  1
No  7   1    Yes 21  1    Yes 5   1
No  16  1    Yes 20  0    Yes 11  1
No  11  0    Yes 5   1    Yes 21  1
;
run;

*****  selects from the ods output tables: Quartiles, SurvDiff ;
*****  selects frou the ods output plots: SurvivalPlot   ;

ods select Quartiles SurvDiff SurvivalPlot ;
proc lifetest data=selection;
time surv_month*cens(0);
strata weightloss/ diff=all test=logrank;
run;
ods select off;




ods trace on ;
proc lifetest data=selection;
time surv_month*cens(0);
strata weightloss/ diff=all test=logrank;
run;
ods trace off;






/****************************************************/
/****************************************************/
*****  problem : adding footnotes to keep track of output


%LET USER = Johnny Awesome ;
%LET CURTIME=%SYSFUNC(TIME(),TIMEAMPM11.0);
%LET CURDATE=%SYSFUNC(TODAY(),WORDDATE.);
FOOTNOTE "THIS OUTPUT WAS CREATED AT &CURTIME ON &CURDATE BY &USER" ;

data selection;
input weightloss $ surv_month   cens @@;
datalines;
No  1   1    No  11  1    Yes 15  1
No  38  0    No  6   0    Yes 13  0
No  9   0    No  9   1    Yes 14  1
No  9   1    No  4   1    Yes 71  0
No  13  1    No  20  1    Yes 37  0
No  20  1    No  44  1    Yes 17  1
No  8   0    No  81  0    Yes 32  0
No  4   1    No  74  0    Yes 17  1
No  27  1    No  13  1    Yes 13  1
No  62  1    Yes 6   1    Yes 27  1
No  12  1    Yes 39  1    Yes 8   1
No  32  1    Yes 100 1    Yes 63  1
No  17  1    Yes 30  0    Yes 22  1
No  15  1    Yes 1   1    Yes 5   1
No  15  1    Yes 2   1    Yes 62  0
No  8   0    Yes 2   0    Yes 9   1
No  22  1    Yes 6   1    Yes 15  1
No  7   1    Yes 21  1    Yes 5   1
No  16  1    Yes 20  0    Yes 11  1
No  11  0    Yes 5   1    Yes 21  1
;
run;

